
Σειρά Ασκήσεων 2: 
Βρόχοι και Επικοινωνία Κονσόλας στον SPIM

Προθεσµία έως ∆ευτέρα 3 Μαρτίου 2008, ώρα 23:59 (βδοµάδα 3.1) (από βδ.2.1)

2.1   Εντολές ∆ιακλάδωσης υπό Συνθήκη στον MIPS

Όπως λέγαµε και στην §1.3, γιά να εκτελεστεί ένα πρόγραµµα, οι εντολές του γράφονται στην κεντρική
µνήµη η µία "κάτω" από την άλλη, δηλαδή σε συνεχόµενες θέσεις (διευθύνσεις) µνήµης. Μετά την
ανάγνωση και εκτέλεση µιάς εντολής, ο επεξεργαστής αυξάνει τον PC κατά το µέγεθος της εντολής που
εκτελέστηκε, οπότε αυτός (ο PC) δείχνει στην επόµενη (την "από κάτω") εντολή. Η σειριακή αυτή
εκτέλεση εντολών διακόπτεται όταν εκτελείται µιά εντολή µεταφοράς ελέγχου (CTI - control transfer
instruction). Είδαµε ήδη µία τέτοια, την εντολή άλµατος j label ("jump" to label), που κάνει ώστε η
επόµενη εντολή που θα εκτελεστεί να είναι η εντολή στη διεύθυνση µνήµης label, αντί να είναι η "από
κάτω" εντολή. Με άλλα λόγια, η εντολή j label φορτώνει τη διεύθυνση label στον καταχωρητή PC.
Χρησιµοποιώντας αυτή την εντολή άλµατος στην άσκηση 1 φτιάξαµε έναν "άπειρο βρόχο", δηλαδή
κάναµε τον υπό προσοµοίωση υπολογιστή να εκτελεί συνεχώς το ίδιο "µπλόκ" εντολών.

Γιά να φτιάξουµε ένα κανονικό (όχι άπειρο) βρόχο χρειαζόµαστε µιάν εντολή διακλάδωσης υπό
συνθήκη (conditional branch), δηλαδή µιάν εντολή που µερικές φορές προκαλεί διακλάδωση και
µερικές φορές όχι, ανάλογα µε το αν ισχύει ή δεν ισχύει κάποια κατάλληλη συνθήκη. Η βασική τέτοια
εντολή είναι η beq (branch if equal): Η εντολή "beq $16, $17, label" διαβάζει τους καταχωρητές
16 και 17, και τους συγκρίνει. Εάν τους βρεί ίσους (equal) διακλαδίζεται στη θέση label, δηλαδή κάνει
τον επεξεργαστή να διαβάσει και εκτελέσει την εντολή από εκείνη τη διεύθυνση σαν επόµενη εντολή.
Αλλοιώς, δεν κάνει τίποτα, οπότε επόµενη εντολή θα διαβαστεί και εκτελεστεί η "από κάτω" εντολή. Η
εντολή bne (branch if not equal) κάνει τα ανάποδα, δηλαδή διακλαδίζεται εάν βρεί τους καταχωρητές
άνισους (not equal), αλλοιώς συνεχίζει "από κάτω".

2.2   Κώδικας Βρόχου και Εισόδου/Εξόδου Κονσόλας

Γιά να επικοινωνούν τα προγράµµατα που τρέχουµε στον SPIM µε τον έξω κόσµο, ο SPIM
προσοµοιώνει µερικές υποτυπώδεις υπηρεσίες λειτουργικού συστήµατος γιά είσοδο/έξοδο (I/O) στην
"κονσόλα" (ένα απλό τερµατικό ASCII). ∆εν είναι ανάγκη προς στιγµήν να καταλάβετε όλες τις
λεπτοµέρειες του πώς γίνεται η κλήση αυτών των λειτουργιών (system call) --αρκεί να µιµηθείτε το
παρακάτω παράδειγµα και να καταλάβετε τις εξηγήσεις που δίνονται κάτω από αυτό. Μελετήστε και
αντιγράψτε σε ένα αρχείο (π.χ. "ask2.s") τον παρακάτω κώδικα --ή διάφορες παραλλαγές του που
προτιµάτε-- και τρέξτε τον στον SPIM.

                # compute s = 1+2+3+...+(n-1),  for n>=2
                # register $16: n
                # register $17: s
                # register $18: i

        .data           # init. data memory with the strings needed:
str_n:  .asciiz "n = "
str_s:  .asciiz "       s = "
str_nl: .asciiz "\n"

        .text           # program memory:
        .globl main             # label "main" must be global;
                                # default trap.handler calls main.
        .globl loop             # global symbols can be specified
                                # symbolically as breakpoints.

main:                       # (1) PRINT A PROMPT:
        addi    $2, $0, 4       # system call code for print_string
        la      $4, str_n       # pseudo-instruction: address of string
        syscall                 # print the string from str_n
                            # (2) READ n (MUST be n>=2 --not checked!):
        addi    $2, $0, 5       # system call code for read_int
        syscall                 # read a line containing an integer
        add     $16, $2, $0     # copy returned int from $2 to n
                            # (3) INITIALIZE s and i:
        add     $17, $0, $0     # s=0;
        addi    $18, $0, 1      # i=1;
loop:                       # (4) LOOP starts here
        add     $17, $17, $18   # s=s+i;
        addi    $18, $18, 1     # i=i+1;
        bne     $18, $16, loop  # repeat while (i!=n)
                            #     LOOP ENDS HERE
                            # (5) PRINT THE ANSWER:
        addi    $2, $0, 4       # system call code for print_string
        la      $4, str_s       # pseudo-instruction: address of string
        syscall                 # print the string from str_s
        addi    $2, $0, 1       # system call code for print_int
        add     $4, $17, $0     # copy argument s to $4
        syscall                 # print the integer in $4 (s)
        addi    $2, $0, 4       # system call code for print_string
        la      $4, str_nl      # pseudo-instruction: address of string
        syscall                 # print a new-line
                            # (6) START ALL OVER AGAIN (infinite loop)
        j       main            # unconditionally jump back to main

Ο κώδικας αυτός υπολογίζει το άθροισµα s=1+2+3+...+(n-1), γιά n µεγαλύτερο ή ίσο του 2 --προσοχή:
αν δοθεί n µικρότερο του 2, ο κώδικας θα µπεί σε (σχεδόν) άπειρο βρόγχο! Η "καρδιά" του κωδικά είναι
τα κοµάτια (3) --αρχικοποιήσεις-- και (4) --βρόγχος υπολογισµού. Προσέξτε τις παρακάτω εξηγήσεις:

Το κοµάτι κάθε γραµµής µετά το # είναι σχόλια, όπως είπαµε και στην άσκηση 1.
Οι γραµµές που αρχίζουν µε τελεία (".") είναι οδηγίες (directives) προς τον Assembler, και όχι
εντολές Assembly του MIPS. Ο πλήρης κατάλογος των οδηγιών που δέχεται ο SPIM βρίσκεται
στις σελίδες Α-51 έως Α-53 του Παραρτήµατος Α, σε περίπτωση που θέλετε να τον
συµβουλευθείτε.
Η οδηγία .data σηµαίνει ότι ό,τι ακολουθεί είναι δεδοµένα (και όχι εντολές), και πρέπει να
τοποθετηθούν στο κοµάτι της µνήµης που προορίζεται γιά αυτά (data segments) (στον SPIM
αυτό αρχίζει από τη διεύθυνση 10000000 δεκαεξαδικό).
Η οδηγία .asciiz σηµαίνει να αρχικοποιήσει ο Assembler τις επόµενες θέσεις (bytes) µνήµης
µε το ASCII string που ακολουθεί, τερµατισµένο µε ένα NULL byte όπως και στην C. Οι
ετικέτες (labels) str_n, str_s, και str_nl, ακολουθούµενες από άνω-κάτω τελεία ":", ορίζουν
την κάθε ετικέτα σαν την διεύθυνση µνήµης όπου ο Assembler βάζει το αντίστοιχο string (τη
διεύθυνση µνήµης του πρώτου byte του string).
Η οδηγία .text σηµαίνει, όπως είπαµε και στην άσκηση 1, ότι ό,τι ακολουθεί είναι εντολές (και
όχι δεδοµένα), και πρέπει να τοποθετηθούν στο κοµάτι της µνήµης που προορίζεται γιά αυτές
(text segments).
Οι οδηγίες .globl λένε στον Assembler να βάλει τις ετικέτες (labels) main και loop στον
πίνακα καθολικών (global) συµβόλων. Γιά την ετικέτα main, είπαµε στην άσκηση 1 γιατί
χρειάζεται αυτό. Γιά την ετικέτα loop (που είναι η αρχή του βρόχου µας), µε το να την κάνουµε
global, µπορούµε να την δίνουµε και συµβολικά --όχι µόνο αριθµητικά-- σαν διεύθυνση
breakpoint στον SPIM.

Το κοµάτι (1) του κώδικα είναι ένα κάλεσµα του λειτουργικού συστήµατος (system call)
προκειµένου να τυπωθεί το string str_n στην κονσόλα (πρόκειται γιά το string "n = " που
ορίστηκε παραπάνω). Γιά να καταλάβει το λειτουργικό σύστηµα ποιό από όλα τα system calls
ζητάµε, βάζουµε στον καταχωρητή $2 σαν παράµετρο (argument) τον αριθµό 4, που σηµαίναι
ότι ζητάµε το system call υπ' αριθµό 4, που είναι το print_string (τα system calls που υλοποιεί ο
SPIM περιγράφονται στις σελίδες Α-48 και Α-49 του Παραρτήµατος Α). Επίσης, γιά να ξέρει το
λειτουργικό σύστηµα ποιό string θέλουµε να τυπώσει στην κονσόλα, βάζουµε στον καταχωρητή
$4 σαν παράµετρο (argument) τη διεύθυνση µνήµης αυτού του string (δηλ. έναν pointer σε αυτό
το string), που στην περίπτωσή µας είναι η ετικέτα str_n που ορίσαµε παραπάνω (το "la" είναι
ψεύδοεντολή (pseudoinstruction) του Assembler του SPIM, και όχι κανονική εντολή του MIPS,

και λέει στον Assembler να γεννήσει µία ή δύο πραγµατικές εντολές που τοποθετούν τη
διεύθυνση της ετικέτας str_n στον καταχωρητή $4, ανάλογα αν η διεύθυνση αυτή χωρά ή όχι
στα 16 bits µιάς σταθεράς "immediate" όπως θα δούµε αργότερα).
Το κοµάτι (2) του κώδικα είναι ένα ανάλογο κάλεσµα (το κάλεσµα υπ' αριθµό 5, δηλαδή
read_int), που περιµένει να διαβάσει έναν ακέραιο από την κονσόλα: ο προσοµοιωτής θα
περιµένει εκεί µέχρι να πληκτρολογήστε έναν ακέραιο και ένα RETURN στο παράθυρο "SPIM
Console". Μέσω της επόµενης εντολής, add, ο ακέραιος που επιστρέφει το κάλεσµα (στον
καταχωρητή $2) αρχικοποιεί τη µεταβλητή µας n (στον καταχωρητή $16).
Το κοµάτι (3) του κώδικα είναι η αρχικοποίηση των µεταβλητών s (καταχωρητής $17) και i
(καταχωρητής $18) πριν µπούµε στο βρόχο.
Το κοµάτι (4) του κώδικα είναι ο κυρίως βρόχος υπολογισµού. Σε κάθε επανάληψή του αυξάνει
το s κατά i και το i κατά 1, και στη συνέχεια συγκρίνει το i (καταχωρητής $18) µε το n
(καταχωρητής $16) και διακλαδίζεται (πηγαίνει) πίσω στην ετικέτα loop, δηλαδή στην αρχή του
βρόχου, όσο αυτές οι δύο µεταβλητές δεν είναι ίσες µεταξύ τους, δηλαδή όσο το i δεν έφτασε
ακόµα το n. Αλλοιώς, µόλις το (ήδη αυξηµένο) i γίνει ίσο µε n, δεν διακλαδιζόµαστε πίσω, αλλά
συνεχίζουµε µε την επόµενη εντολή, δηλαδή το κοµάτι (5) του κώδικα.
Το κοµάτι (5) του κώδικα είναι τρία καλέσµατα συστήµατος γιά να τυπωθούν το string str_s, η
απάντηση s, και το string str_nl. Τέλος, η εντολή jump στο (6) µας επιστρέφει πάντα πίσω στο
main, ώστε το πρόγραµµα να ξανατρέχει συνεχώς µέχρι να τερµατίσετε τον SPIM.

Άσκηση 2.3:   Τρέξιµο στον SPIM

Ξεκινήστε το xspim µε τον τρόπο που είπαµε στην §1.4, και φορτώστε το αρχείο µε το
παραπάνω πρόγραµµα που γράψατε µέσω του κουµπιού "load".
Μέσω του κουµπιού "step" ζητήστε single-stepping, δηλαδή να εκτελούνται µιά-µιά οι εντολές
και να τις βλέπετε. Η εκτέλεση αρχίζει στη διεύθυνση __start (0x00400000) όπου υπάρχει
κώδικας από το αρχείο trap.handler. Στη διέυθυνση 0x00400014 υπάρχει µιά εντολή καλέσµατος
διαδικασίας (jal --jump and link) η οποία καλεί τον κώδικά σας στο main, και η εκτέλεση
πηγαίνει στη διεύθυνση 0x00400020. Όταν φτάστε στο δεύτερο κάλεσµα συστήµατος
(0x00400030), µην ξεχάσετε να πληκτρολογήσετε έναν ακέραιο µεγαλύτερο ή ίσο του 2 (πρέπει
να τον πληκτρολογήσετε αφού ο SPIM φτάσει εκεί --παλαιότερες πληκτρολογήσεις συνήθως
χάνονται).
Αφού βαρεθείτε να βλέπετε τις ενολές να εκτελούνται µία-µία, χρησιµοποιήστε το κουµπί
"breakpoints" γιά να ορίσετε ένα breakpoint στο πρόγραµµά σας, π.χ. τη διεύθυνση "loop".
Μετά, πείτε στο πρόγραµµα να τρέξει, µέσω του κουµπιού "run", οπότε αυτό τρέχει "σιωπηλά"
µέχρι να ξαναφτάσει στη διεύθυνση loop. Έτσι µπορείτε να επιταχύνετε την παρακολούθηση
ενός προγράµµατος, και να το κάνετε να σταµατάει σε "ενδιαφέροντα" ή "ύποπτα" σηµεία.
Τέλος, αφαιρέστε όλα τα breakpoints και τρέξτε το πρόγραµµα κανονικά, οπότε θα βλέπετε µόνο
τις εισόδους και εξόδους στην κονσόλα.

Τρόπος Παράδοσης: 
Θα παραδώσετε ηλεκτρονικά ένα στιγµιότυπο της οθόνης καθώς τρέχετε το πρόγραµµα "xspim" και
αυτό βρίσκεται σ' ένα "ενδιαφέρον" ενδιάµεσο breakpoint. Το στιγµιότυπο θα το πάρετε και θα το
παραδώσετε κατ' αναλογία πρός την παράδοση της άσκησης 1:

Σε µηχανή LINUX/X-Windows: χρησιµοποιήστε το πρόγραµµα "xv" και την επιλογή του
"Grab". Σώστε την εικόνα σε µορφή jpeg (.jpg), σε αρχείο µε το όνοµα "ask2.jpg".
Σε µηχανή PC/Windows: πατήστε το κουµπί "Print Screen" του πληκτρολογίου, και µετά
ανοίξτε το Microsoft Photo Editor και κάνετε "paste". Σώστε την εικόνα αυτή σε µορφή Jpeg
(.jpg), και µεταφέρετε το αρχείο σε µηχανή SPARC µε το όνοµα "ask2.jpg".
Θα παραδώσετε την άσκησή σας µάλλον µε τη "νέα" διαδικασία παράδοσης, µάλλον µέσω της
ιστοσελίδας www.csd.uoc.gr/~hy225/submit/ (ο παλαιός τρόπος παράδοσης ήταν να εκτελέστε
"~hy225/bin/submit 2", από το directory στο οποίο βρίσκεται το αρχείο σας "ask2.jpg" (µε
αυτό και µόνο το όνοµα)· η εντολή submit δουλεύει µόνο σε πλατφόρµα SPARC/Sunos).

